F818 FOR A/D EXPERIMENTS

For experiments involving analog to digital conversion, it would be convenient to have an F84 look-alike with one A/D channel. We will make one using a PIC16F818. In the bargain, we will get two other features:

- Internal clock-oscillator which may be used, or not.
- In-Circuit debugging capability.

PINS AND FUNCTIONS



The methods for selecting pin functions are summarized in the following table:

PIC16F818 I/O

RA0/AN0	Always I/O	Select via software using ADCON1 register
RA1/AN1		
RA2/AN2/Vref-		
RA3/AN3/Vref+		
RA4/AN4/T0CKI		
CCP1	CCP1 on pin RB2 or RB3	Select via config word
RA5/$\overline{\text{MCLR}}$/Vpp	Digital Input vs. $\overline{\text{MCLR}}$	
RA6/OSC2/CLKOUT	Digital I/O vs. OSC2 vs. clock output	
RA7/OSC1/CLKIN	Digital I/O vs. OSC1 vs. clock input	

The ADCON1 register is used to select a single analog pin/channel for our examples.

ADFM	ADCS2	—	—	PCFG3	PCFG2	PCFG1	PCFG0	0x9F
7							0	

Power-on reset 00000000

There are fifteen possible combinations of pin assignments. We will use AN0 and make the remaining pins digital I/O.

PCFG

	AN4	AN3	AN2	AN1	AN0
1110	D	D	D	D	A

D = Digital
A = Analog

CLOCK OSCILLATOR

The internal clock oscillator may be used (most common) in this update, or an external clock oscillator may be used. The details of various external oscillator circuits and components as well as internal clock oscillator use options are given in the Microchip data sheet. At device programming time, the part must be told via configuration bits which clock oscillator option will be used.

When the internal clock-oscillator has been selected via the configuration register, the frequency is selected using three bits in the OSCCON register.


```
Bits 6,5,4    IRCF2:IRCF1:IRCF0: Internal oscillator
                   frequency selection bits
              111 = 8 MHz
              110 = 4 Mhz
              101 = 2 Mhz
              100 = 1 MHz
              011 = 500 KHz
              010 = 250 KHz
              001 = 125 KHz
              000 = 31.25 KHz (default)

Bit 2         IOFS: INTOSC frequency stable bit
```

For our experiments, we will write `b'01100000'` to the OSCCON oscillator control register to select 4 MHz as the oscillator frequency.

ARCHITECTURE - OVERVIEW

Program Memory

The PIC16F818 program memory is 14 bits wide and 1K words long. Program memory is flash.



Since the F818 program memory is all on one page, paging is not an issue.

File Registers

The file registers are 8 bits wide with the exception of the PCLATH register which is 5 bits wide. The file registers are divided into four banks. A simplified file register map which shows the most commonly used file registers follows.



CONFIGURATION BITS

The configuration bits and what they control are shown in the following diagram:

CP	CCPMX	DEBUG	WRT1	WRT0	CPD	LVP	BOREN	MCLRE	FOSC2	$\overline{\text{PWRTEN}}$	WDTEN	FOSC1	FOSC0
13													0

Bit 13 **CP:** Flash program memory code protection bit
1 = Code protection off
0 = Code protection on

Bit 12 **CCPMX:** CCP1 pin selection bit
1 = CCP1 function on RB2
0 = CCP1 function on RB3

Bit 11 **DEBUG:** In-Circuit Debugger mode bit
1 = In-Circuit Debugger disabled, RB7 and RB6 are I/O pins
0 = In-Circuit Debugger enabled, RB7 and RB6 are
dedicated to the debugger

Bits 10,9 **WRT1:WRT0:** Flash program memory write enable bits
11 = Write protection off
10 = 0x000-0x01FF write-protected, 0x0200 to 0x03ff
may be modified by EECON control
01 = 0x000-0x03FF write-protected

Bit 8 **CPD:** Data EE memory code protection bit
1 = Code protection off
0 = Code protection on

Bit 7 **LVP:** Low voltage in-circuit serial programming enable bit
1 = Enabled - RB3 is programming pin
0 = Disabled - RB3 is digital I/O pin

Bit 6 **BOREN:** Brown-out reset enable bit (1)
1 = Enabled
0 = Disabled

Bit 5 **MCLRE:** RA5/MCLR pin function select bit
1 = MCLR
0 = RB5 is digital input pin

Bit 3 **PWRTEN:** Power-up timer enable bit (1)
1 = Disabled
0 = Enabled

Bit 2 **WDTEN:** Watchdog timer enable bit
1 = Enabled
0 = Disabled

Bits 4,1,0 **FOSC2:FOSC1:FOSC0:** Oscillator selection bits
111 = EXTRC oscillator - CLKOUT on RA6 pin
110 = EXTRC oscillator - I/O on RA6 pin
101 = INTRC oscillator - CLKOUT on RA6, I/O on RA7
100 = INTRC oscillator - I/O on RA6, RA7 pins
011 = EXTCLK - I/O on RA6 pin
010 = HS oscillator - high speed crystal/resonator
001 = XT oscillator - crystal/resonator
000 = LP oscillator - low power crystal

For our examples, we will use the internal oscillator. The configuration word is determined as follows:



10-bit A/D example code for the F818 is presented below.

In-circuit debugging using a PICkit 2 or PICkit 3 is outlined in another Update available on this web site.

CIRCUIT FOR EXPERIMENTS



SAMPLE A/D CODE

10-bit A/D example code for the F818 follows:

```
;=======F818AD.ASM=========================2/6/11==
        list    p=16f818
        __config   h'3f70'
        radix   hex
;-----------------------------------------------------
;       cpu equates (memory map)
status  equ     0x03
porta   equ     0x05
portb   equ     0x06
intcon  equ     0x0b
adresh  equ     0x1e
adcon0  equ     0x1f
trisa   equ     0x85
trisb   equ     0x86
osccon  equ     0x8f
adresl  equ     0x9e
adcon1  equ     0x9f
count   equ     0x20
ncount  equ     0x21
mcount  equ     0x22
temp    equ     0x23
;-----------------------------------------------------
;       bit equates
rp0     equ     5
;-----------------------------------------------------
        org     0x000
start   bsf     status,rp0  ;switch to bank 1
        movlw   b'01100000' ;4 MHz clock oscillator
        movwf   osccon
        movlw   b'00000001' ;inputs/outputs
        movwf   trisa
        movlw   b'00000000' ;inputs/outputs
        movwf   trisb
        movlw   b'00001110' ;port A, bit 0 analog input
        movwf   adcon1      ;   bits 4,3,2,1 digital I/O,
;                               result left justified
        bcf     status,rp0  ;switch back to bank 0
        clrf    portb       ;LEDs off
        bcf     porta,1
        bcf     porta,4
        bcf     intcon,7    ;global interrupt disable
meas    movlw   b'10000001' ;configure A/D - select AN0
        movwf   adcon0      ;   select conv clock, AN0 on
        call    del_20      ;delay 20 microseconds
        bsf     adcon0,2    ;start conversion
test    btfsc   adcon0,2    ;test go/done bit
        goto    test
        movf    adresh,w    ;conv complete, get A/D result
;                               ms 8 bits
```

```
        movwf   portb       ;display ms 8 data bits via LEDS
        bsf     status,rp0  ;switch to bank 1
        movf    adresl,w    ;get A/D result ls 2 bits
        bcf     status,rp0  ;switch to bank 0
        movwf   temp        ;store for disection
        btfss   temp,7
        goto    clr_4
        bsf     porta,4     ;display via LED
do_0    btfss   temp,6
        goto    clr_1
        bsf     porta,1     ;display via LED
wait    call    debounce    ;wait a while (200 milliseconds)
        goto    meas        ;look at voltage again
;-------------------------------------------------------------
clr_4   bcf     porta,4     ;display via LED
        goto    do_0
clr_1   bcf     porta,1     ;display via LED
        goto    wait
;-------------------------------------------------------------
del_20  movlw   0x07        ;delay 20 microseconds
        movwf   count
repeat  decfsz  count
        goto    repeat
        return
;-------------------------------------------------------------
debounce  movlw 0xff        ;M
        movwf   mcount      ;to M counter
loadn   movlw   0xff        ;N
        movwf   ncount      ;to N counter
decn    decfsz  ncount,f    ;decrement N
        goto    decn        ;again
        decfsz  mcount,f    ;decrement M
        goto    loadn       ;again
        return              ;done
;-------------------------------------------------------------
        end
;-------------------------------------------------------------
;at device programming time, select:
;       memory unprotected
;       watchdog timer disabled (default is enabled)
;       internal clock oscillator, I/O on port B, pins 7 & 6
;       mclr active
;       power-up timer on
;       brown-out reset enabled
;       lvp disabled
;       debug mode disabled
;       CCP1 on on port B, pin 2 (not used)
;=============================================================
```